

## 제10강

# 시리얼통신 인터럽트

관련 SFR

송수신 루틴

동작모드

보레이트 설정

시리얼 ISR

실습과제

Ref.) Chapter 10

## 관련 SFR

### \* SBUF(Serial Buffer) SFR

- : 송신할 혹은 수신한 데이터를 일시적으로 저장하기 위한 SFR
- : 송신 및 수신에서 공통으로 사용
- : 하지만, 송신 SBUF, 수신 SBUF가 별도 존재(전이중 전송가)

송신시)

MOV SBUF,A

혹은, MOV SBUF,#0FEH

수신시)

MOV A,SBUF

## 관련 SFR(계속)

### \* **SCON(Serial Control) SFR**

: 바이트 및 비트 단위 접근 가능

7	6	5	4	3	2	1	0
<b>SM0</b>	<b>SM1</b>	SM2	<b>REN</b>	TB8	RB8	<b>TI</b>	<b>RI</b>

SM0, SM1 : 모드 설정(4종)

REN(Receive Enable) : 수신 가능 여부(1:가능, 0:불가능)

; 수신회로의 스타트/스톱 회로를 리셋하고 수신 가능 역할

TI : 송신완료시 하드웨어적으로 set(**송신가능상태**)

RI : 수신완료시 하드웨어적으로 set(**수신완료상태**)

\* 초기상태 : TI=1, RI=0

\* TI, RI는 재송수신을 위해서 **소프트웨어적으로 clear해야 함**

## 관련 SFR(계속)

### \* PCON(Power Control) SFR

: 바이트 단위 접근만 가능

7	6	5	4	3	2	1	0
SMOD	—	—	—	GF1	GF0	PD	IDL

:시리얼 클럭 발생회로는 T/C1과 PCON의 SMOD비트 이용

: SMOD=1이면 두배의 보레이트 생성

모드	SMOD='0'	SMOD='1'
0	사용안함	사용안함
1	T/C1 오버플로의 1/2	T/C1 오버플로
2	XTAL의 1/4	XTAL의 1/2
3	T/C1 오버플로의 1/2	T/C1 오버플로

사용예) MOV PCON,#10000000B ; 바이트단위접근

## 송수신시 루틴

### \* 수신 루틴

LOOP:

CLR RI	;
JNB RI,\$	; 수신완료시까지 대기
MOV A,SBUF	; 우선 읽어들이고 후, RI clear
JMP LOOP	

### \* 송신 루틴

LOOP:

CLR TI	;
MOV SBUF,A	;
JNB TI,\$	; 송신완료시까지 대기
JMP LOOP	

## 동작 모드

### \* 모드 0

: I/O 확장용의 동기식 통신 모드

: Shift Register 방식 이용

: TxD(P3.0)으로 동기 신호(12분주),  
RxD(P3.1)으로 데이터 송수신

### \* 모드 1

: 10비트 프레임의 보레이트 가변 비동기식 통신 모드

Start	D0	D1	D2	D3	D4	D5	D6	D7	Stop
-------	----	----	----	----	----	----	----	----	------

## 동작 모드(계속)

### \* 모드 2

: 11비트 프레임의 보레이트 고정 비동기식 통신 모드

Start	D0	D1	D2	D3	D4	D5	D6	D7	TB8	Stop
-------	----	----	----	----	----	----	----	----	-----	------

### \* 모드 3

: 11비트 프레임의 보레이트 가변 비동기식 통신 모드

Start	D0	D1	D2	D3	D4	D5	D6	D7	TB8	Stop
-------	----	----	----	----	----	----	----	----	-----	------

### \* 비동기식

: 1bit의 검출은 12클럭중 7,8,9 클럭 체크 2회이상 동일신호!

## 보레이트(baud rate) 설정

- \* T/C1의 모드 2(8비트 자동 재적재) 이용 경우  
=> TH1의 시상수

PCON.7(SMOD)=0인 경우 :

$$TH1 = 256 - \frac{Crystal}{384 \times Baud}$$

PCON.7(SMOD)=1인 경우 :

$$TH1 = 256 - \frac{Crystal}{192 \times Baud}$$



## 보레이트(baud rate) 설정(계속)

예) 11.059Mhz XTAL, 19200 보레이트 생성하려면 TH1 = ?

PCON.7(SMOD)=0인 경우,

$$TH1 = 256 - \frac{11059000}{384 \times 19200} = 256 - \frac{28799}{19200} = 256 - 1.5 = 254.5$$

TH1 < -254 => 보레이트 14400,

TH1 < -255 => 보레이트 28800 => 정확한 생성불가

PCON.7(SMOD)=1인 경우,

$$TH1 = 256 - \frac{11059000}{192 \times 19200} = 256 - \frac{57699}{19200} = 256 - 3 = 253$$

**253 => (0FDH)**

## 보레이트(baud rate) 설정(계속)

\* 대부분, T/C1의 Mode 2(Auto Reload) 사용하여  
9600bps로 통신 => 실습키트 19200!!

보레이트	시스템클럭	SMOD	T/C 1		
			$C/\overline{T}$	Mode	TH1
Mode0 Max : 1M	12MHz	x	x	x	x
Mode2 Max : 375K	12MHz	1	x	x	x
Mode1,3 : 62.5K	12MHz	1	0	2	FFH
<b>19.2K</b>	<b>11.0592MHz</b>	<b>1</b>	<b>0</b>	<b>2</b>	<b>FDH</b>
9.6K	11.0592MHz	0	0	2	FDH
4.8K	11.0592MHz	0	0	2	FAH
2.4K	11.0592MHz	0	0	2	F4H
1.2K	11.0592MHz	0	0	2	E8H
137.5	11.0592MHz	0	0	2	1DH
110	6MHz	0	0	2	72H
110	12MHz	0	0	1	FEEDH

## 시리얼 ISR

**\* 송신 및 수신에 대해 통합된 시리얼 인터럽트 요청**

**=> ISR에서 송신인지 수신인지 판별해야 함**

```
SERIAL_ISR:
RX_ISR:      JNB RI,TX_ISR          ; 수신인지 판별
              MOV A,SBUF            ; RI 서비스
              CLR RI                ;
TX_ISR:      JNB TI,RTN             ; 송신인지 판별
              CLR TI                ; TI 서비스
              MOV SBUF,#40H
RTN:          RETI
```

**\* TI, RI는 재송수신을 위해서 소프트웨어적으로 clear해야 함**

## 예제 실습

### [실습1] 문자 단위의 반송 프로그램(인터럽트방식)

: 수신후, 수신을 확인후 반송토록 구현( ISR 들락임 )

```

1      ;=====
2      ;   SERIAL_01.ASM
3      ;       Char ECHO, Interrupt
4      ;=====
5      $mod51
8000      6      ORG 8000H
8000 803E      7      SJMP START
           8      ;
8023      9      ORG 8023H
8023 803C     10     SJMP SERIAL_ISR
           11     ;
8040      12     ORG 8040H
8040 75813F   13     START: MOV SP,#3FH
8043 C2D5     14     CLR PSW.5           ; 수신시 1
8045 114A     15     ACALL SERIAL_INIT
  
```

```


16          ;
8047 00      17      LOOP:  NOP          ; normal
8048 80FD      18          SJMP LOOP
19          ;
804A          20      SERIAL_INIT:
804A C2AF      21          CLR EA
804C 758920    22          MOV TMOD,#00100000B      ; 00100000B
804F 758780    23          MOV PCON,#80H          ; SMOD
8052 758DFD    24          MOV TH1,#0FDH
8055 D28E      25          SETB TR1
26          ;
8057 759852    27          MOV SCON,#01010010B      ; 01010010B
805A D2BC      28          SETB PS
805C D2AC      29          SETB ES
805E D2AF      30          SETB EA
8060 22        31          RET
32          ;
8061          33      SERIAL_ISR:      ; interrupt service routine
8061          34      RX_ISR:
8061 30980B    35          JNB RI,TX_ISR
8064 C2AF      36          CLR EA
8066 E599      37          MOV A,SBUF

```

8068 C298	38	CLR RI	
806A D2D5	39	SETB PSW.5	
806C D2AF	40	SETB EA	
806E 32	41	RETI	
	42	;	
806F	43	TX_ISR:	
806F 10D501	44	JBC PSW.5, TX_CHR	; 수신여부
8072 32	45	RETI	
	46	;	
8073	47	TX_CHR:	; 직전수신시만
8073 C2AF	48	CLR EA	
8075 C299	49	CLR TI	
8077 F599	50	MOV SBUF, A	
8079 D2AF	51	SETB EA	
807B 32	52	RETI	
	53	;	
	54	END	

**보완) 보다 단순하게 처리 가능(수신하면 곧바로 송신토록)**

## 참고) '속성'-'ASCII설정'



## 예제실습(계속)

### [실습2] 문자 단위의 반송 프로그램(폴링방식)

	1	;=====
	2	; SERIAL_02.ASM
	3	; Char ECHO, Polling
	4	;=====
	5	\$mod51
8000	6	ORG 8000H
8000 75813F	7	MOV SP,#3FH
	8	;
8003 110B	9	ACALL SERIAL_INIT
8005 111F	10	LOOP: ACALL RX_CHR
8007 1127	11	ACALL TX_CHR
8009 80FA	12	SJMP LOOP
	13	;
800B	14	SERIAL_INIT:
800B C2AF	15	CLR EA
800D 758920	16	MOV TMOD,#00100000B



## 마이크로프로세서및실습 시리얼통신인터럽트 17

8010 758780	17	MOV PCON,#80H	; SMOD bit
8013 758DFD	18	MOV TH1,#0FDH	
8016 D28E	19	SETB TR1	
	20	;	
8018 759850	21	MOV SCON,#01010000B	
801B 759900	22	MOV SBUF,#0	
801E 22	23	RET	
	24	;	
801F	25	RX_CHR:	
801F 3098FD	26	JNB RI,\$	
8022 E599	27	MOV A,SBUF	
8024 C298	28	CLR RI	
8026 22	29	RET	
	30	;	
8027	31	TX_CHR:	
8027 3099FD	32	JNB TI,\$	
802A C299	33	CLR TI	
802C F599	34	MOV SBUF,A	
802E 22	35	RET	
	36	;	
	37	END	

## 예제실습(계속)

### [실습3] 문자열 반송(폴링방식)

: 한 라인의 문자열 단위로 반송하는 프로그램

: Enter key = CR + LF

```

1      ;=====
2      ;   SERIAL_03.ASM
3      ;       LINE ECHO, Polling
4      ;=====
5      $mod51
6      ;
000D   7      CR      EQU 0DH
000A   8      LF      EQU 0AH
8100   9      ALINE   EQU 8100H      ; 수신문자열
10     ;
8000   11     ORG 8000H
8000 75813F 12     MOV SP,#3FH
13     ;
8003 110F   14     ACALL SERIAL_INIT

```

8005 1123	15	LOOP:	ACALL RX_LINE	
8007 113B	16		ACALL TX_LF	; line feed
8009 1130	17		ACALL TX_LINE	
800B 113B	18		ACALL TX_LF	; line feed
800D 80F6	19		SJMP LOOP	
	20		;	
800F	21	SERIAL_INIT:		
800F C2AF	22		CLR EA	
8011 758920	23		MOV TMOD,#00100000B	
8014 758780	24		MOV PCON,#80H	; SMOD bit
8017 758DFD	25		MOV TH1,#0FDH	
801A D28E	26		SETB TR1	
	27		;	
801C 759850	28		MOV SCON,#01010000B	
801F 759900	29		MOV SBUF,#0	
8022 22	30		RET	
	31		;	
8023	32	RX_LINE:		
8023 908100	33		MOV DPTR,#ALINE	
8026	34	RX_LOOP:		
8026 1140	35		ACALL RX_CHR	
8028 F0	36		MOVX @DPTR,A	

8029 A3	37	INC DPTR
802A 1148	38	ACALL TX_CHR ; echo
802C B40DF7	39	CJNE A,#CR,RX_LOOP
802F 22	40	RET
	41	;
8030	42	TX_LINE:
8030 908100	43	MOV DPTR,#ALINE
8033	44	TX_LOOP:
8033 E0	45	MOVX A,@DPTR
8034 1148	46	ACALL TX_CHR
8036 A3	47	INC DPTR
8037 B40DF9	48	CJNE A,#CR,TX_LOOP
803A 22	49	RET
	50	;
803B	51	TX_LF:
803B 740A	52	MOV A,#LF
803D 1148	53	ACALL TX_CHR
803F 22	54	RET
	55	;
8040	56	RX_CHR:
8040 3098FD	57	JNB RI,\$
8043 E599	58	MOV A,SBUF

8045 C298	59	CLR RI
8047 22	60	RET
	61	;
8048	62	TX_CHR:
8048 3099FD	63	JNB TI,\$
804B C299	64	CLR TI
804D F599	65	MOV SBUF,A
804F 22	66	RET
	67	;
	68	END

## 실습과제

### [과제1] 문자반송

: [실습1]예제 참조

: +2한 문자 반송

### [과제2] 문자열 반송(인터럽트방식)

: 한 라인의 문자열 단위로 반송하는 프로그램

: [실습3]을 인터럽트 방식으로 재구현

## 실습과제(계속)

### [과제3] 문자열 반송

: 여러 문자열 준비후, 요청문자열의 전송

### [과제4] LED 원격제어

: 전송문자열이 !로 시작될 때, 뒤따르는 문자(0..F)들을  
수치값으로 하여 LED에 출력  
(예, !3A45F )

### [\*] 11장 CLCD 예습하기

: 한번쯤 읽어 올 것!!!